﻿<!--#include file="h.asp"--><%
'
'	七色虎建站系统
'	网友投稿页面
'	2011.2.1


Sub main
	'投稿配置
	If bbsEssay<>"1" And bbsEssay<>"0" Then bbsEssay="1"'会员投稿
	
	checkLogin()'验证登录
	checkUserStatus()'检测用户是否在服刑
	Dim a,userid,text,title,count,body_
	hu.title = "会员投稿"
	body_ = "[div=nav]-"&hu.url("?aid=index","首页")&"-"&hu.url("bbs_user.asp","会员中心")&"-会员投稿[/div]"
	a=getN("a", 0)
	userid=user_array(0)
	count=hu.conn("select count(id) from 74hu_article where classid=0 and hu_author='"&userid&"'")(0)
	If bbsEssay=0 Then body_ = body_ & "[div=tip]注:管理员已关闭投稿功能[/div]"
	If a=0 Then
		If count>0 Then body_ = body_ & "已有"&count&"篇稿子未通过审核<br/>"
		Dim formarray(3)
		formarray(0) = http_path&"bbs_essay.asp?a=1||post||发布投稿||标题:<br/>"
		formarray(1) = "input||title$$$$20$$30||<br/>内容:<br/>"
		formarray(2) = "area||text$$$$20||<br/>"
		formarray(3) = "hidden||"&bbsSid&"$$"&sid&"||"
		body_ = body_ & hu.getForm(formarray) & "<br/>注:投稿内容需管理员审核"
	Else
		If bbsEssay=0 Then rupt "错误", "管理员已关闭投稿功能，请与管理员联系"
		title=getD("title","")
		text=getD("text","")
		If text="" Or title="" Then rupt "错误", "标题或内容不能为空"
		If len(text)<50 Then rupt "错误", "投稿内容不能低于50字"
		If ifArticleRepeat(title) Then rupt "错误", "站内已存在同样标题的文章"
		hu.conn("insert into 74hu_article (hu_date,title,test,hu_author,classid)values('"&time_now&"','"&title&"','"&text&"',"&userid&",0)")
		body_ = body_ & "投稿成功,请等待管理员审核<br/>"
	End If
	hu.body = comUbb(body_)
End Sub

%>